Subject: adding a colorbar legend to a contour plot? Posted by haferman on Thu, 10 Oct 1996 07:00:00 GMT

View Forum Message <> Reply to Message

I've got datapoints (x,y,z) and I'm using IDL to produce a contour plot of z at locations (x,y).

This is easy enough to do with IDL, but I don't see in the manuals (we have the Version 3.6 manuals) how I can add a colorbar legen to the contour plots.

I'm new to this newsgroup; is there a FAQ, or could somone mail me a small snippet of code to perform this task?

Thanks.

Jeff

Subject: Re: adding a colorbar legend to a contour plot? Posted by davidf on Thu, 10 Oct 1996 07:00:00 GMT

View Forum Message <> Reply to Message

Jeff Haferman <a href="mailto:haferman@audry.gsfc.nasa.gov">haferman@audry.gsfc.nasa.gov</a> writes:

- > I've got datapoints (x,y,z) and I'm using IDL
- > to produce a contour plot of z at locations (x,y).

\_

- > This is easy enough to do with IDL, but I don't see
- > in the manuals (we have the Version 3.6 manuals) how
- > I can add a colorbar legend to the contour plots.

Suppose you wanted to add a horizontal color bar over the top of your contour plot. Here is one way to do it.

First, load some colors. Let's suppose you want 200 colors.

```
LOADCT, 5, NCOLORS=200 ncolors = 200
```

Next, leave some space at the top of your contour plot for the color bar with the POSITION keyword. Like this:

```
CONTOUR, DIST(31, 41), NLEVELS=25, $
C_COLORS=INDGEN(25)*8, POSITON=[0.15, 0.15, 0.95, 0.8]
```

Now, suppose you want the color bar to be the length of the

contour plot, and positioned above it. You might specify its location like this:

```
loc = [0.15, 0.80, 0.95, 0.90]
```

I like to place my color bars in NORMAL coordinates, so that they go into any sized window. (Most of my graphics windows are resizeable these days.)

Now create a color bar. (For a vertical bar exchange the two expressions on either side of the matrix multiplier.)

```
bar = BINDGEN(256) # REPLICATE(1B, 10)
```

Size the color bar for the current graphics window.

```
xsize = (loc(2) - loc(0)) * !D.X_VSIZE
ysize = (loc(3) - loc(1)) * !D.Y_VSIZE
bar = CONGRID(bar, CEIL(xsize), CEIL(ysize), /INTERP)
```

Scale the color bar to the number of colors you are using.

```
bar = BYTSCL(bar, TOP=ncolors-1)
```

Display the color bar in the window.

```
TV, bar, loc(0), loc(1), /NORMAL
```

Draw a box around the bar.

```
PLOTS, [loc(0), loc(0), loc(2), loc(2), loc(0)], $ [loc(1), loc(3), loc(3), loc(1), loc(1)], /NORMAL
```

Now, you can add annotations to the bar as needed. Many people use XYOUTS. I like to use the PLOT command to label the color bar because this gives me tick marks, nice labels, etc. It's up to you.

You can download a fancier version of this from my anonymous ftp site. Use anonymous ftp to the machine ftp.frii.com. Look in /pub/dfanning/outgoing/idl\_examples for the file colorbar.pro.

Note that this is only \*one\* way to produce color bars in IDL. You will probably discover other code and ways if you poke around at the various IDL sites.

Hope this gives you some ideas.

## Cheers!

David

--

David Fanning, Ph.D. Phone: 970-221-0438 Fax: 970-221-4728

E-Mail: davidf@fortnet.org

Subject: Re: adding a colorbar legend to a contour plot?
Posted by on Mon, 14 Oct 1996 07:00:00 GMT

View Forum Message <> Reply to Message

## Jeff Haferman wrote:

>

- > I've got datapoints (x,y,z) and I'm using IDL
- > to produce a contour plot of z at locations (x,y).

>

- > This is easy enough to do with IDL, but I don't see
- > in the manuals (we have the Version 3.6 manuals) how
- > I can add a colorbar legen to the contour plots.

>

- > I'm new to this newsgroup; is there a FAQ, or could
- > somone mail me a small snippet of code to perform this
- > task?

>

> Thanks,

>

> Jeff

Hi Jeff,

I have the same problem of making a color-coded temperature map of a 2-dimmensional area. For this purpose I have writen a routine making a contour plot overlaying a color image ov the data (TVSCL). On the levt side a color bar is plotted. If you like you could have a copy of this. It is pretty much independend and works even in a !p.multi environment. Besides you could have four x,y axes, postscript plots,..... But simply looking at the code will make you understand the way of producing color bars. Unfortunately it is not thoroughly documented, but you will understand.

Gunter

--

Kargl G�nter Mailto: kargl@limar1.mpae.gwdg.de

Max-Planck Inst, f. Aeronomie Phone: ++49 5556 979 234

Max-Planck Str. 2 Fax:: ++49 5556 979 240

D- 37191 Katlenburg-Lindau

Germany